require u-boot-phytium-sdk.inc

SUMMARY = "U-Boot bootloader fw_printenv/setenv utilities"
DEPENDS += "mtd-utils"

PROVIDES = "u-boot-fw-utils"
RPROVIDES:${PN} = "u-boot-fw-utils"

ENV_CONFIG_FILE:pe2201 = "fw_env_evb-pe2201.config"
SRC_URI += "file://${ENV_CONFIG_FILE}"

INSANE_SKIP:${PN} = "already-stripped"
EXTRA_OEMAKE:class-target = ' \
    CROSS_COMPILE=${TARGET_PREFIX} CC="${CC} ${CFLAGS} ${LDFLAGS}" \
    HOSTCC="${BUILD_CC} ${BUILD_CFLAGS} ${BUILD_LDFLAGS}" V=1 \
    '
EXTRA_OEMAKE:class-cross = 'HOSTCC="${CC} ${CFLAGS} ${LDFLAGS}" V=1'

do_compile() {
    oe_runmake -C ${S} O=${B} ${UBOOT_MACHINE}
    oe_runmake envtools
}

do_install() {
    install -d ${D}${base_sbindir}
    install -m 755 ${B}/tools/env/fw_printenv ${D}${base_sbindir}/fw_printenv
    ln -sf fw_printenv ${D}${base_sbindir}/fw_setenv

    install -d ${D}${sysconfdir}
    install -m 644 ${WORKDIR}/${ENV_CONFIG_FILE} ${D}${sysconfdir}/fw_env.config
}

do_install:class-cross() {
    install -d ${D}${bindir_cross}
    install -m 755 ${B}/tools/env/fw_printenv ${D}${bindir_cross}/fw_printenv
    ln -sf fw_printenv ${D}${bindir_cross}/fw_setenv
}

FILES:${PN} = "${base_sbindir}/fw_printenv ${base_sbindir}/fw_setenv ${sysconfdir}/fw_env.config"
FILES:class-cross = "${bindir_cross}/fw_printenv ${bindir_cross}/fw_setenv"

SYSROOT_DIRS:append:class-cross = " ${bindir_cross}"

PACKAGE_ARCH = "${MACHINE_ARCH}"
BBCLASSEXTEND = "cross"
